



## **Progetto di Ricerca Operativa**

*Ottimizzazione dei Costi per i Servizi di Trasporto Ordinario  
e di Emergenza*

Soldà Matteo  
Matricola 1226319

A.A. 2022–2023

# Contents

<b>1</b>	<b>Introduzione</b>	<b>3</b>
1.1	Abstract . . . . .	3
1.2	Problema Generale . . . . .	3
<b>2</b>	<b>Modello</b>	<b>4</b>
2.1	Insiemi . . . . .	4
2.2	Parametri . . . . .	4
2.3	Variabili . . . . .	4
2.4	Funzione Obiettivo . . . . .	4
<b>3</b>	<b>Codice AMPL</b>	<b>6</b>
3.1	File <i>.mod</i> . . . . .	6
<b>4</b>	<b>Scenari</b>	<b>8</b>
4.1	Primo Scenario . . . . .	8
4.1.1	Descrizione . . . . .	8
4.1.2	Dati . . . . .	8
4.1.3	Output . . . . .	8
4.2	Secondo Scenario . . . . .	8
4.2.1	Descrizione . . . . .	8
4.2.2	Dati . . . . .	8
4.2.3	Output . . . . .	9
4.3	Terzo Scenario . . . . .	9
4.3.1	Descrizione . . . . .	9
4.3.2	Output . . . . .	9
<b>5</b>	<b>Extra</b>	<b>10</b>
5.1	Output su File . . . . .	10
5.2	Stampa Personalizzata per Problema Non Risolvibile . . . . .	10

# 1 Introduzione

## 1.1 Abstract

La Centrale Operativa del 118 della città di Padova necessita di un gran numero di ambulanze per svolgere i trasporti sia ordinari che di emergenza nell'intera provincia, ma con le ambulanze interne e quelle in possesso degli ospedali periferici si riesce a coprire solo in minima parte il fabbisogno: si rende quindi necessario rivolgersi alle Organizzazioni di Volontariato che, in cambio di un rimborso spese, prestano ambulanze e soccorritori per lo svolgimento degli interventi. L'obiettivo del progetto è quello di minimizzare il costo per il mantenimento del servizio, garantendo comunque un pronto intervento in caso di necessità.

## 1.2 Problema Generale

Ogni giorno la Centrale Operativa del 118 di Padova e l'Azienda Ospedaliera ricevono una moltitudine di chiamate, sia per richieste di supporto immediato (che vengono gestite dal SUEM), sia per trasporti programmati (che passano per le linee dell'Azienda Ospedaliera e che non rappresentano casi di emergenza, ma dedicati al trasporto di persone invalide presso l'ospedale per le visite programmate oppure per le dimissioni dal Pronto Soccorso).

L'Azienda Ospedaliera, al suo interno, dispone di un determinato numero di ambulanze di tipo B (dedicate ai trasporti), mentre quelle di tipo A (dedicate al SUEM) sono fornite dagli ospedali periferici. L'insieme di questi mezzi non è però sufficiente a coprire le richieste pervenute dall'intera provincia, ma ne ricopre una minima parte. Per questo motivo, la Centrale Operativa si rivolge a tre Organizzazioni di Volontariato (OdV) che offrono il loro supporto per coprire i viaggi scoperti:

- La Croce Rossa può fornire solo ambulanze di tipo A
- La Croce Bianca può fornire solo ambulanze di tipo B
- La Croce Verde può fornire sia ambulanze di tipo A che di tipo B

Il numero di ambulanze complessive in un determinato giorno viene stimato rispetto alla media di interventi e trasporti effettuati negli stessi giorni delle settimane precedenti, mantenendo comunque un numero arbitrario di ambulanze sia di tipo A che di tipo B in più.

Si vuole quindi minimizzare il costo complessivo settimanale per il servizio, sapendo che:

- Ogni giorno la Centrale Operativa ha necessità di un quantitativo diverso di ambulanze
- Qualora si attivasse una OdV per l'assistenza settimanale, è prevista una singola quota fissa di attivazione
- Ogni ambulanza richiamata dalle OdV prevede un rimborso spese giornaliero che varia in base all'organizzazione e al tipo di ambulanza attivata
- Le ambulanze dell'Azienda Ospedaliera e degli ospedali periferici, se attivate in un determinato giorno, prevedono dei piccoli costi di manutenzione
- Ogni OdV e ospedale dispone di un numero limitato di ambulanze
- Ogni giorno bisogna attivare un numero arbitrario di ambulanze in più rispetto al fabbisogno

**Disclaimer:** questo progetto riguarda un tema reale con dati fittizi. Le Organizzazioni di Volontariato citate nell'introduzione e l'Ospedale di Padova sono realmente esistenti, ma i dati riguardanti disponibilità di mezzi, stima della necessità, metodologie di intervento e costi sostenuti sono casuali e adattati per rendere consistente lo studio in questione. Ogni riferimento è quindi da ritenersi puramente casuale in quanto frutto dell'immaginazione del redattore.

## 2 Modello

### 2.1 Insiemi

- *Giorni*: giorni della settimana in cui è attivo il servizio
- *FornitoriA*: fornitori che dispongono di ambulanze di tipo A da attivare
- *FornitoriB*: fornitori che dispongono di ambulanze di tipo B da attivare

### 2.2 Parametri

- *bisognoA<sub>g</sub>*: fabbisogno per il giorno *g* di ambulanze di tipo A (escluso surplus arbitrario)
- *bisognoB<sub>g</sub>*: fabbisogno per il giorno *g* di ambulanze di tipo B (escluso surplus arbitrario)
- *surplusA<sub>g</sub>*: surplus di ambulanze di tipo A per il giorno *g*
- *surplusB<sub>g</sub>*: surplus di ambulanze di tipo B per il giorno *g*
- *maxA<sub>fa</sub>*: numero massimo di ambulanze di tipo A che il fornitore *fa* può fornire in un giorno
- *maxB<sub>fb</sub>*: numero massimo di ambulanze di tipo B che il fornitore *fb* può fornire in un giorno
- *costoGiornalieroA<sub>fa</sub>*: costo per l'attivazione giornaliera per una singola ambulanza di tipo A del fornitore *fa*
- *costoGiornalieroB<sub>fb</sub>*: costo per l'attivazione giornaliera per una singola ambulanza di tipo B del fornitore *fb*
- *costoAttivazioneA<sub>fa</sub>*: costo settimanale per l'attivazione di un fornitore *fa* per la fornitura settimanale di ambulanze di tipo A
- *costoAttivazioneB<sub>fb</sub>*: costo settimanale per l'attivazione di un fornitore *fb* per la fornitura settimanale di ambulanze di tipo B
- *BigM*: utilizzato per vincolare l'utilizzo delle ambulanze rispetto all'attivazione del fornitore (costante sufficientemente grande). Questo, nei file **.dat** è stato definito utilizzando la più vicina potenza del 2 rispetto a  $Max(bisognoA, bisognoB) + Max(surplusA, surplusB)$ .

### 2.3 Variabili

- *ambulanzeA<sub>fa,ga</sub>* = numero di ambulanze del fornitore *fa* attivate il giorno *ga*
- *ambulanzeB<sub>fb,gb</sub>* = numero di ambulanze del fornitore *fb* attivate il giorno *gb*
- $attivazioneSettimanaleA_{fa} = \begin{cases} 1 & \text{se viene attivato il fornitore di ambulanze di tipo A } fa \text{ per la settimana} \\ 0 & \text{altrimenti} \end{cases}$
- $attivazioneSettimanaleB_{fb} = \begin{cases} 1 & \text{se viene attivato il fornitore di ambulanze di tipo B } fb \text{ per la settimana} \\ 0 & \text{altrimenti} \end{cases}$

### 2.4 Funzione Obiettivo

La funzione obiettivo del problema può essere espressa in forma generale come segue:

$$\begin{aligned} \min \quad & \text{Costo Giornaliero Ambulanze Tipo A} + \text{Costo Giornaliero Ambulanze Tipo B} \\ & + \text{Costo Attivazione Ambulanze Tipo A} + \text{Costo Attivazione Ambulanze Tipo B} \end{aligned}$$

Dove i termini sono così definiti:

$$\text{Costo Giornaliero Ambulanze Tipo A} = \sum_{f \in \text{FornitoriA}, g \in \text{Giorni}} \text{ambulanzeA}[f, g] \cdot \text{costoGiornalieroA}[f]$$

$$\text{Costo Giornaliero Ambulanze Tipo B} = \sum_{f \in \text{FornitoriB}, g \in \text{Giorni}} \text{ambulanzeB}[f, g] \cdot \text{costoGiornalieroB}[f]$$

$$\text{Costo Attivazione Ambulanze Tipo A} = \sum_{f \in \text{FornitoriA}} \text{attivazioneSettimanaleA}[f] \cdot \text{costoAttivazioneA}[f]$$

$$\text{Costo Attivazione Ambulanze Tipo B} = \sum_{f \in \text{FornitoriB}} \text{attivazioneSettimanaleB}[f] \cdot \text{costoAttivazioneB}[f]$$

subject to

- Ogni giorno la Centrale Operativa ha necessità di un numero diverso di ambulanze:

$$\sum_{f \in \text{FornitoriA}} \text{ambulanzeA}[f, g] \geq \text{bisognoA}[g] + \text{surplusA}[g] \quad \forall g \in \text{Giorni}$$

$$\sum_{f \in \text{FornitoriB}} \text{ambulanzeB}[f, g] \geq \text{bisognoB}[g] + \text{surplusB}[g] \quad \forall g \in \text{Giorni}$$

- I fornitori dispongono di un numero massimo di ambulanze fornibili in un giorno:

$$\text{ambulanze}[f, g] \leq \text{maxA}[f] \quad \forall f \in \text{FornitoriA}, g \in \text{Giorni}$$

$$\text{ambulanze}[f, g] \leq \text{maxB}[f] \quad \forall f \in \text{FornitoriB}, g \in \text{Giorni}$$

- Le ambulanze di un determinato fornitore si possono attivare se e solo se si è pagata la quota di attivazione settimanale:

$$\text{ambulanzeA}[f, g] \leq \text{BigM} \cdot \text{attivazioneSettimanaleA}[f]$$

$$\text{ambulanzeB}[f, g] \leq \text{BigM} \cdot \text{attivazioneSettimanaleB}[f]$$

**Domini:**

- $\text{ambulanzeA}_{f,g}, \text{ambulanzeB}_{f,g} \geq 0$
- $\text{attivazioneSettimanaleA}_f, \text{attivazioneSettimanaleB}_f \in 0, 1$

## 3 Codice AMPL

### 3.1 File *.mod*

---

```
# =====
#      Modello AMPL
#      Ricerca operativa
#      Solda' Matteo
#      (2022)
# =====

### INSIEMI ###
set Giorni ordered; # Giorni della settimana
set FornitoriA;    # Fornitori di ambulanze di tipo A
set FornitoriB;    # Fornitori di ambulanze di tipo B

### PARAMETRI ###
# Fabbisogno Giornaliero di Ambulanze
param bisognoA { Giorni } >= 0 integer default 0;
param bisognoB { Giorni } >= 0 integer default 0;

# Ambulanze "di Scorta"
param surplusA { Giorni } >= 2 integer default 2;
param surplusB { Giorni } >= 1 integer default 1;

# Numero Massimo Giornaliero di Ambulanze Attiviabili da un Fornitore
param maxA { FornitoriA } >= 0 integer default 15;
param maxB { FornitoriB } >= 0 integer default 5;

# Costo per l'Attivazione Giornaliero di una Ambulanza
param costoGiornalieroA { FornitoriA } > 0 default 10;
param costoGiornalieroB { FornitoriB } > 0 default 5;

# Costo per l'Attivazione Settimanale di una Ambulanza (da pagare una sola
# volta in caso di attivazione del fornitore per la determinata settimana)
param costoAttivazioneA { FornitoriA } >= 0 default 100;
param costoAttivazioneB { FornitoriB } >= 0 default 75;

# BigM per vincoli di tipo logico
param BigM >= 0 integer default 500;

### VARIABILI ###
var ambulanzeA { fa in FornitoriA, ga in Giorni } integer >= 0; # Numero
# di ambulanze di tipo A fornite dal fornitore fa il giorno ga
var ambulanzeB { fb in FornitoriB, gb in Giorni } integer >= 0; # Numero
# di ambulanze di tipo B fornite dal fornitore fb il giorno gb
var attivazioneSettimanaleA { f in FornitoriA } binary; # Variabile logica
# per l'attivazione settimanale delle ambulanze di tipo A di un certo
# fornitore
var attivazioneSettimanaleB { f in FornitoriB } binary; # Variabile logica
# per l'attivazione settimanale delle ambulanze di tipo B di un certo
# fornitore

### FUNZIONE OBIETTIVO ###
minimize costo:
    (sum { f in FornitoriA, g in Giorni } ambulanzeA[f, g] *
      costoGiornalieroA[f]) +      # COSTO SETTIMANALE AMBULANZE TIPO A
```

```

    (sum { f in FornitoriB, g in Giorni } ambulanzeB[f, g] *
      costoGiornalieroB[f]) +      # COSTO SETTIMANALE AMBULANZE TIPO B
    (sum { f in FornitoriA } attivazioneSettimanaleA[f] *
      costoAttivazioneA[f]) +      # ATTIVAZIONE SETTIMANALE AMBULANZE TIPO B
    (sum { f in FornitoriB } attivazioneSettimanaleB[f] *
      costoAttivazioneB[f])        # ATTIVAZIONE SETTIMANALE AMBULANZE TIPO B
  ;

### VINCOLI ###
# Vincoli Necessita' Giornaliera
subject to necessitaGiornalieraA {g in Giorni} : sum { f in FornitoriA }
  ambulanzeA[f, g] >= bisognoA[g] + surplusA[g];
subject to necessitaGiornalieraB {g in Giorni} : sum { f in FornitoriB }
  ambulanzeB[f, g] >= bisognoB[g] + surplusB[g];

# Vincoli Disponibilita' Fornitori
subject to disponibilitaA { f in FornitoriA, g in Giorni } : ambulanzeA[f,
  g] <= maxA[f];
subject to disponibilitaB { f in FornitoriB, g in Giorni } : ambulanzeB[f,
  g] <= maxB[f];

# Vincoli Logici
subject to attivazioneA { f in FornitoriA, g in Giorni } : ambulanzeA[f,
  g] <= BigM * attivazioneSettimanaleA[f];
subject to attivazioneB { f in FornitoriB, g in Giorni } : ambulanzeB[f,
  g] <= BigM * attivazioneSettimanaleB[f];

```

---

## 4 Scenari

**Premessa:** dato che negli scenari proposti gli insiemi non variano, questi saranno di seguito presentati una sola volta. Gli insiemi sono così definiti:

Giorni	lunedì	martedì	mercoledì	giovedì	venerdì	sabato	domenica
--------	--------	---------	-----------	---------	---------	--------	----------

FornitoriA	rossa	verde	interna
------------	-------	-------	---------

FornitoriA	bianca	verde	interna
------------	--------	-------	---------

### 4.1 Primo Scenario

#### 4.1.1 Descrizione

In questo primo scenario, i dati sono stati inseriti manualmente, con un numero di ambulanze necessarie tale da permettere di trovare una soluzione ottima.

#### 4.1.2 Dati

BisognoA		BisognoB		surplusA		surplusB	
lunedì	20	lunedì	3	lunedì	3	lunedì	1
martedì	22	martedì	5	martedì	3	martedì	3
mercoledì	10	mercoledì	2	mercoledì	3	mercoledì	1
giovedì	20	giovedì	3	giovedì	2	giovedì	1
venerdì	30	venerdì	4	venerdì	5	venerdì	2
sabato	30	sabato	2	sabato	6	sabato	1
domenica	15	domenica	0	domenica	4	domenica	1

maxA		maxB		costoGiornalieroA		costoGiornalieroB		costoAttivazioneA	
rossa	15	bianca	5	rossa	15	bianca	20	rossa	130
verde	30	verde	10	verde	18	verde	12	verde	50
interna	3	interna	8	interna	8	interna	5	interna	0

costoAttivazioneB	
bianca	70
verde	55
interna	0

#### 4.1.3 Output

L'esecuzione del file **.run** mostra che il costo minimo è di €2929.

### 4.2 Secondo Scenario

#### 4.2.1 Descrizione

Il secondo set di dati differisce dal primo solo per la richiesta di ambulanze sia di tipo A che di tipo B per il lunedì che non è soddisfacibile. Per questo motivo, nel prossimo paragrafo saranno riportati solo i parametri modificati.

#### 4.2.2 Dati

BisognoA		BisognoB	
lunedì	150	lunedì	65
martedì	22	martedì	5
mercoledì	10	mercoledì	2
giovedì	20	giovedì	3
venerdì	30	venerdì	4
sabato	30	sabato	2
domenica	15	domenica	0

maxA		maxB	
rossa	15	bianca	5
verde	30	verde	10
interna	3	interna	8



### 4.2.3 Output

Ovviamente, il costo minimo trovato eseguendo il file **.run** sarà pari a €0.

## 4.3 Terzo Scenario

### 4.3.1 Descrizione

Nel terzo scenario sono stati definiti soltanto gli insiemi, così da poter sfruttare i valori di default. Per questo motivo non verrà riportato il valore dei vari parametri in questo specificato nel modello.

### 4.3.2 Output

Eseguendo il file **.run** dedicato, utilizzando quindi i dati di default, il costo minimo risulta essere di €350.

## 5 Extra

### 5.1 Output su File

All'interno della cartella **AMPL** si può trovare un file denominato *Ospedale.GenerateOutputFile.run* che esegue in serie tutti e tre gli scenari sopracitati e, tramite un secondo file *.run* di supporto denominato *GenerateOutput.run*, ne salva l'output in un file dedicato chiamato *Output.txt*. Per fare ciò, sono stati utilizzati gli operatori `>` e `>>`. Il primo serve a creare un nuovo file (o a ripulirlo se già esistente) e scriverci l'istruzione della stessa linea; il secondo serve invece per aprire il file e fare un *append* dell'istruzione di riferimento.

### 5.2 Stampa Personalizzata per Problema Non Risolvibile

All'interno dei file *Ospedale.xxx.run* dove *xxx* indica il numero del caso preso in esame o il file dedicato alla stampa, c'è un *if-then-else statement* che, utilizzando il valore della variabile di stato *solve\_result*, determina se un problema è risolvibile o meno e modifica l'output di conseguenza. Nel caso il problema non fosse risolvibile, tramite la variabile *solve\_result\_num*, indica il codice che identifica la non risolvibilità del problema.